
1. Overview

This document lists all changes needed to correct problems in the TTCN implementation of test case 13.4.4.1 which is part of the LTE/SAE test suite ‘iwd-EUTRA-B2012-03_D12wk48’.
The test case can be demonstrated to run with one LTE UE (see section 5). Execution log is provided as evidence.

3. Verification Test Summary

Test Case:
TC_13_4_4_1
Test Group:
13\MultiLayer_Procedures_EC2K.ttcn
ATS Version:
iwd-EUTRA-B2012-03_D12wk48
System Simulator used:
R&S CMW500 
UE used:
LG Electronics Optimus G
Verification Status:
PASS


4. Corrections required for test case 13.4.4.1
This section describes the TTCN changes required to make test case 13.4.4.1 run correctly with a LTE UE. The TTCN used to pass this test case belongs to ‘iwd-EUTRA-B2012-03_D12wk48’ release.

Change 1 – Corrections to function ‘f_TC_13_4_4_1_CDMA2000()’
	Function name
	f_TC_13_4_4_1_CDMA2000()

	Reason for change
	1. Incorrect cell number of CDMA2000 cells in comment line.

2. Missing type parameter in the Initialization function. 

3. Coordination message was missing at the end of the test body.


	Summary of change
	1. Replaced “cell15” with “cell19 and cell21”.

2. Added parameter “type1XRTT”.

3. Added a step to send Coordination trigger to the EUTRA side.


	TTCN module
	13\MultiLayer_Procedures_EC2K_CDMA2000.ttcn

	MCC160 Comment
	


Before change

…
   //****************************************************************
  /*
   * @desc      testcase 13.4.4.1:
   *            Pre-registration at 1xRTT and Cell reselection / 1x Zone Registration
   * @status
   */
   function f_TC_13_4_4_1_CDMA2000 () runs on CDMA2000_PTC
  {
    ///Initialise test case
    f_CDMA2000_Init ( EUTRA_CDMA2000 );
    //Send Sys info for ARFCN
    f_CDMA2000_Send_IR_SysInfoToEUTRA (cdma2000_Cell19 );
     f_CDMA2000_Send_IR_SysInfoToEUTRA (cdma2000_Cell21 );
    //Create CDMA2000 cell15 and broadcast System Infomation
    f_CDMA2000CreateCell ( cdma2000_Cell19 );
    f_CDMA2000CreateCell ( cdma2000_Cell21 );
    f_CDMA2000_PreRegister( cdma2000_Cell19,  px_RAND );
    //First Trigger: to start test case
    f_IRAT_WaitForCoOrd_Trigger (EUTRA);
    f_CDMA2000_TestBody_Set( true );
    //------------------------------------------------
    //Start TestBody
    //@siclog "Step 3" siclog@
    //Second Trigger: to Change cell power according to T1
    f_IRAT_WaitForCoOrd_Trigger (EUTRA);
    f_CDMA2000_RegisterRTT1XCell( cdma2000_Cell21, true, px_RAND2);
    f_IRAT_SendCoOrd ( EUTRA, cms_IRAT_Trigger );
    //End TestBody
    //------------------------------------------------
    f_CDMA2000_TestBody_Set ( false );
    //  release cells
    f_CDMA2000_ReleaseCells();
  }; //end of testcase 13.4.4.1
…
After change

…
   //****************************************************************
  /*
   * @desc      testcase 13.4.4.1:
   *            Pre-registration at 1xRTT and Cell reselection / 1x Zone Registration
   * @status
   */
   function f_TC_13_4_4_1_CDMA2000 () runs on CDMA2000_PTC
  {
    ///Initialise test case
    f_CDMA2000_Init ( EUTRA_CDMA2000, type1xRTT );
    //Send Sys info for ARFCN
    f_CDMA2000_Send_IR_SysInfoToEUTRA (cdma2000_Cell19 );
     f_CDMA2000_Send_IR_SysInfoToEUTRA (cdma2000_Cell21 );
    //Create CDMA2000 cell19 and cell21 and broadcast System Infomation
    f_CDMA2000CreateCell ( cdma2000_Cell19 );
    f_CDMA2000CreateCell ( cdma2000_Cell21 );
    f_CDMA2000_PreRegister( cdma2000_Cell19,  px_RAND );
    //First Trigger: to start test case
    f_IRAT_WaitForCoOrd_Trigger (EUTRA);
    f_CDMA2000_TestBody_Set( true );
    //------------------------------------------------
    //Start TestBody
    //@siclog "Step 3" siclog@
    //Second Trigger: to Change cell power according to T1
    f_IRAT_WaitForCoOrd_Trigger (EUTRA);
    f_CDMA2000_RegisterRTT1XCell( cdma2000_Cell21, true, px_RAND2);
    f_IRAT_SendCoOrd ( EUTRA, cms_IRAT_Trigger );
    //End TestBody
    //------------------------------------------------
    f_CDMA2000_TestBody_Set ( false );

    //send coordination message for end of test case
    f_IRAT_SendCoOrd (EUTRA, cms_IRAT_Trigger);
    //  release cells
    f_CDMA2000_ReleaseCells();
  }; //end of testcase 13.4.4.1
…
Change 2 – Corrections to template ‘car_SRB2_RrcPdu_IND()’
	Template name
	car_SRB2_RrcPdu_IND()

	Reason for change
	In this receive template a PDU is expected on SRB2 and not on SRB1.

	Summary of change
	Corrected the test suite constant to “tsc_SRB2”.

	TTCN module
	CommonEUTRA\EUTRA_CDMA2000Tunneling.ttcn

	MCC160 Comment
	


Before change

…
  //----------------------------------------------------------------------------
  // SRB2: ASP to receive RRC message (
  template SRB_COMMON_IND car_SRB2_RrcPdu_IND(EUTRA_CellId_Type p_CellId,
                                              template (present) UL_DCCH_Message p_RrcPdu,
                                              template (present) TimingInfo_Type p_TimingInfo := ?) :=
  { /* @status     */
    Common := cr_IndAspCommonPart_SRB(p_CellId, tsc_SRB1, p_TimingInfo),
    Signalling := {
      Rrc := {
        Dcch := p_RrcPdu
      },
      Nas := omit
    }
  };
…
After change

…
  //----------------------------------------------------------------------------
  // SRB2: ASP to receive RRC message (
  template SRB_COMMON_IND car_SRB2_RrcPdu_IND(EUTRA_CellId_Type p_CellId,
                                              template (present) UL_DCCH_Message p_RrcPdu,
                                              template (present) TimingInfo_Type p_TimingInfo := ?) :=
  { /* @status     */
    Common := cr_IndAspCommonPart_SRB(p_CellId, tsc_SRB2, p_TimingInfo),
    Signalling := {
      Rrc := {
        Dcch := p_RrcPdu
      },
      Nas := omit
    }
  };    

…
Change 3 – Corrections to function ‘f_CDMA2000_PreRegistration()’
	Function name
	f_CDMA2000_PreRegistration()

	Reason for change
	Function ‘f_CDMA2000_PreRegistration()’ is a generic function for 1xRTT as well as for HRPD applications. Therefore, variable “v_CDMA2000Type” should be used instead of the hardcoded value “type1XRTT” in C2KTUNNEL.send().

	Summary of change
	Replaced the hardcoded value “type1XRTT” variable “v_CDMA2000Type” in the send template cas_CDMA2000_MSG_REQ().

	TTCN module
	CommonEUTRA\EUTRA_CDMA2000Tunneling.ttcn

	MCC160 Comment
	


Before change

…
  //============================================================================
  function f_CDMA2000_PreRegistration(EUTRA_CellId_Type p_CellId) runs on EUTRA_PTC
  { /* NOTE: implementation requires that the SS can cope with subsequent CDMA2000_MSG_REQ
     *       (=> we don't need to wait for the response before we can send another request) */
    var SRB_COMMON_IND v_SRB_COMMON_IND;
    var CDMA2000_MSG_IND v_CDMA2000_MSG_IND;
    var CDMA2000_Type v_CDMA2000Type;
    var ULInformationTransfer_DedicatedInfo_Type v_DedicatedInfoUL;
    var template (value) DLInformationTransfer_DedicatedInfo_Type v_DedicatedInfoDL;
    var DedicatedInfoCDMA2000 v_DedicatedInfoCDMA2000;
    var boolean v_EndOfProcedure := false;
    while (not v_EndOfProcedure) {
      alt {
        [] SRB.receive(car_SRB2_RrcPdu_IND(p_CellId, (cr_ULInformationTransfer_1XRTT, cr_ULInformationTransfer_HRPD))) -> value v_SRB_COMMON_IND {
          v_DedicatedInfoUL := v_SRB_COMMON_IND.Signalling.Rrc.Dcch.message_.c1.ulInformationTransfer.criticalExtensions.c1.ulInformationTransfer_r8.dedicatedInfoType;
          if (ischosen(v_DedicatedInfoUL.dedicatedInfoCDMA2000_1XRTT)) {
            v_CDMA2000Type := type1XRTT;
            v_DedicatedInfoCDMA2000 := v_DedicatedInfoUL.dedicatedInfoCDMA2000_1XRTT;
          } else {
            v_CDMA2000Type := typeHRPD;
            v_DedicatedInfoCDMA2000 := v_DedicatedInfoUL.dedicatedInfoCDMA2000_HRPD;
          }
          C2KTUNNEL.send(cas_CDMA2000_MSG_REQ(type1XRTT, v_DedicatedInfoCDMA2000));
        }
        [] C2KTUNNEL.receive(car_CDMA2000_MSG_IND) -> value v_CDMA2000_MSG_IND {
          v_CDMA2000Type := v_CDMA2000_MSG_IND.CDMA2000Type;
          v_DedicatedInfoCDMA2000 := v_CDMA2000_MSG_IND.DL_Msg.Msg;
          select (v_CDMA2000Type) {
            case (type1XRTT) { v_DedicatedInfoDL := cs_DedicatedInfoType_1XRTT(v_DedicatedInfoCDMA2000); }
            case (typeHRPD)  { v_DedicatedInfoDL := cs_DedicatedInfoType_HRPD(v_DedicatedInfoCDMA2000); }
          }
          SRB.send(cas_SRB2_RrcPdu_REQ(p_CellId, cs_TimingInfo_Now, cs_DLInformationTransfer_Common(tsc_RRC_TI_Def, v_DedicatedInfoDL)));
        }
        [] a_IRAT_WaitForCoOrd_Trigger(CDMA2000) {   /* get trigger from the C2K PTC that pre-registration has been finished */
          v_EndOfProcedure := true;
        }
      }
    }
  }
…
After change

…
  //============================================================================
  function f_CDMA2000_PreRegistration(EUTRA_CellId_Type p_CellId) runs on EUTRA_PTC
  { /* NOTE: implementation requires that the SS can cope with subsequent CDMA2000_MSG_REQ
     *       (=> we don't need to wait for the response before we can send another request) */
    var SRB_COMMON_IND v_SRB_COMMON_IND;
    var CDMA2000_MSG_IND v_CDMA2000_MSG_IND;
    var CDMA2000_Type v_CDMA2000Type;
    var ULInformationTransfer_DedicatedInfo_Type v_DedicatedInfoUL;
    var template (value) DLInformationTransfer_DedicatedInfo_Type v_DedicatedInfoDL;
    var DedicatedInfoCDMA2000 v_DedicatedInfoCDMA2000;
    var boolean v_EndOfProcedure := false;
    while (not v_EndOfProcedure) {
      alt {
        [] SRB.receive(car_SRB2_RrcPdu_IND(p_CellId, (cr_ULInformationTransfer_1XRTT, cr_ULInformationTransfer_HRPD))) -> value v_SRB_COMMON_IND {
          v_DedicatedInfoUL := v_SRB_COMMON_IND.Signalling.Rrc.Dcch.message_.c1.ulInformationTransfer.criticalExtensions.c1.ulInformationTransfer_r8.dedicatedInfoType;
          if (ischosen(v_DedicatedInfoUL.dedicatedInfoCDMA2000_1XRTT)) {
            v_CDMA2000Type := type1XRTT;
            v_DedicatedInfoCDMA2000 := v_DedicatedInfoUL.dedicatedInfoCDMA2000_1XRTT;
          } else {
            v_CDMA2000Type := typeHRPD;
            v_DedicatedInfoCDMA2000 := v_DedicatedInfoUL.dedicatedInfoCDMA2000_HRPD;
          }
          C2KTUNNEL.send(cas_CDMA2000_MSG_REQ(v_CDMA2000Type, v_DedicatedInfoCDMA2000));
        }
        [] C2KTUNNEL.receive(car_CDMA2000_MSG_IND) -> value v_CDMA2000_MSG_IND {
          v_CDMA2000Type := v_CDMA2000_MSG_IND.CDMA2000Type;
          v_DedicatedInfoCDMA2000 := v_CDMA2000_MSG_IND.DL_Msg.Msg;
          select (v_CDMA2000Type) {
            case (type1XRTT) { v_DedicatedInfoDL := cs_DedicatedInfoType_1XRTT(v_DedicatedInfoCDMA2000); }
            case (typeHRPD)  { v_DedicatedInfoDL := cs_DedicatedInfoType_HRPD(v_DedicatedInfoCDMA2000); }
          }
          SRB.send(cas_SRB2_RrcPdu_REQ(p_CellId, cs_TimingInfo_Now, cs_DLInformationTransfer_Common(tsc_RRC_TI_Def, v_DedicatedInfoDL)));
        }
        [] a_IRAT_WaitForCoOrd_Trigger(CDMA2000) {   /* get trigger from the C2K PTC that pre-registration has been finished */
          v_EndOfProcedure := true;
        }
      }
    }
  }
…
Change 4 – Introduction of new function ‘f_EUTRA_IdleUpdated_Step16_17_PreRegistration()’
	Function name
	f_EUTRA_IdleUpdated_Step16_17_PreRegistration()

	Reason for change
	In function “f_EUTRA_NAS_AttachComplete()” the function “f_EUTRA_DelayForUserPlaneSignalling()” is called to handle IMS capable UEs. However, in case of Pre-registration it is indicated in SIB 8 that the UE shall do Pre-registration before user plane signalling.
See also change 5.

	Summary of change
	Added a new generic function for the Pre-Registration case. I.e. to handle first the Pre-registration and after that the user plane signalling.
See also change 5.

	TTCN module
	CommonEUTRA\EUTRA_CommonProcedures.ttcn

	MCC160 Comment
	


Before change

…
  //----------------------------------------------------------------------------
  /*
   * @desc      step 15 .. 17 of UE Registration acc. to 36.508 cl. 4.5.2
   * @param     p_CellId
   * @param     p_ReleaseConnection (default value: true) - default value is to release the connection at the end (i.e. complete step 17) @sic R5-105005 sic@
   * @status    APPROVED
   */
  function f_EUTRA_IdleUpdated_Step16_17(EUTRA_CellId_Type p_CellId,
                                         boolean p_ReleaseConnection := true) runs on EUTRA_PTC
  { /* @sic R5-112600: always delay of 1.2s (i.e. p_WaitForIMS := waitForIMS when calling f_EUTRA_NAS_AttachComplete) sic@ */
    var HalfOctet_Type v_EpsBearerId := tsc_EpsDefaultBearerId;
    f_EUTRA_NAS_AttachComplete(p_CellId, v_EpsBearerId); // step 16  @sic R5s110449 sic@
    if (p_ReleaseConnection) { // @sic R5-105005 Some tests only want to do up to step 16 sic@
      f_EUTRA_RRC_ConnectionRelease(p_CellId);                                  // step 17
    }
  }
…
After change

…
  //----------------------------------------------------------------------------
  /*
   * @desc      step 15 .. 17 of UE Registration acc. to 36.508 cl. 4.5.2
   * @param     p_CellId
   * @param     p_ReleaseConnection (default value: true) - default value is to release the connection at the end (i.e. complete step 17) @sic R5-105005 sic@
   * @status    APPROVED
   */
  function f_EUTRA_IdleUpdated_Step16_17(EUTRA_CellId_Type p_CellId,
                                         boolean p_ReleaseConnection := true) runs on EUTRA_PTC
  { /* @sic R5-112600: always delay of 1.2s (i.e. p_WaitForIMS := waitForIMS when calling f_EUTRA_NAS_AttachComplete) sic@ */
    var HalfOctet_Type v_EpsBearerId := tsc_EpsDefaultBearerId;
    f_EUTRA_NAS_AttachComplete(p_CellId, v_EpsBearerId); // step 16  @sic R5s110449 sic@
    if (p_ReleaseConnection) { // @sic R5-105005 Some tests only want to do up to step 16 sic@
      f_EUTRA_RRC_ConnectionRelease(p_CellId);                                  // step 17
    }
  }
  function f_EUTRA_IdleUpdated_Step16_17_PreRegistration(EUTRA_CellId_Type p_CellId) runs on EUTRA_PTC
  { /* @sic R5-112600: always delay of 1.2s (i.e. p_WaitForIMS := waitForIMS when calling f_EUTRA_NAS_AttachComplete) sic@ */
    var HalfOctet_Type v_EpsBearerId := tsc_EpsDefaultBearerId;
    f_EUTRA_NAS_AttachComplete_PreRegistration(p_CellId, v_EpsBearerId); // step 16  @sic R5s110449 sic@
  }
…
Change 5 – Introduction of new function ‘f_EUTRA_NAS_AttachComplete_PreRegistration()’
	Function name
	f_EUTRA_NAS_AttachComplete_PreRegistration()

	Reason for change
	See also change 4 above.

	Summary of change
	Added a new generic function for the Pre-Registration case. 
See also change 4 above.

	TTCN module
	CommonEUTRA\EUTRA_NASSteps.ttcn

	MCC160 Comment
	


Before change

…
  //----------------------------------------------------------------------------
  /*
   * @desc      receive ATTACH COMPLETE from SRB2; ATTACH COMPLETE contains piggybacked ACTIVATE DEFAULT EPS BEARER CONTEXT ACCEPT (within ESM Message Container)
   * @param     p_CellId
   * @param     p_EpsBearerId
   * @param     p_WaitForIMS        (default value: waitForIMS)
   * @status    APPROVED
   */
  function f_EUTRA_NAS_AttachComplete(EUTRA_CellId_Type p_CellId,
                                      HalfOctet_Type p_EpsBearerId,
                                      EUTRA_WaitForIMS_Type p_WaitForIMS := waitForIMS) runs on EUTRA_PTC
  {
    SRB.receive(car_SRB2_NasPdu_IND(p_CellId,
                                    cr_NAS_IndicationWithPiggybacking(tsc_SHT_IntegrityProtected_Ciphered,
                                                                      cr_508_ATTACH_COMPLETE,
                                                                      cr_508_ActivateDefEpsBearerContextAccept(p_EpsBearerId))));
    f_EUTRA_DelayForUserPlaneSignalling(p_WaitForIMS);   // @sic R5s110449 sic@
  };
…
After change

…
  //----------------------------------------------------------------------------
  /*
   * @desc      receive ATTACH COMPLETE from SRB2; ATTACH COMPLETE contains piggybacked ACTIVATE DEFAULT EPS BEARER CONTEXT ACCEPT (within ESM Message Container)
   * @param     p_CellId
   * @param     p_EpsBearerId
   * @param     p_WaitForIMS        (default value: waitForIMS)
   * @status    APPROVED
   */
  function f_EUTRA_NAS_AttachComplete(EUTRA_CellId_Type p_CellId,
                                      HalfOctet_Type p_EpsBearerId,
                                      EUTRA_WaitForIMS_Type p_WaitForIMS := waitForIMS) runs on EUTRA_PTC
  {
    SRB.receive(car_SRB2_NasPdu_IND(p_CellId,
                                    cr_NAS_IndicationWithPiggybacking(tsc_SHT_IntegrityProtected_Ciphered,
                                                                      cr_508_ATTACH_COMPLETE,
                                                                      cr_508_ActivateDefEpsBearerContextAccept(p_EpsBearerId))));
    f_EUTRA_DelayForUserPlaneSignalling(p_WaitForIMS);   // @sic R5s110449 sic@
  };
  function f_EUTRA_NAS_AttachComplete_PreRegistration(EUTRA_CellId_Type p_CellId,
                                      HalfOctet_Type p_EpsBearerId) runs on EUTRA_PTC
  {
    SRB.receive(car_SRB2_NasPdu_IND(p_CellId,
                                    cr_NAS_IndicationWithPiggybacking(tsc_SHT_IntegrityProtected_Ciphered,
                                                                      cr_508_ATTACH_COMPLETE,
                                                                      cr_508_ActivateDefEpsBearerContextAccept(p_EpsBearerId))));
  };
…
5. Execution Log Files

5.1 LG Electronics Optimus G UE 

The LG Electronics Optimus G UE passed this test case on R&S CMW500 with NAS and AS Snow 3G integrity and ciphering algorithm in LTE FDD band 11 and C2K BC 6. The documentation below is enclosed as evidence of the successful test case run [1]:

· Test Case Execution log files: 
tc_13_4_4_1_LG-Electronics_FDD.log   
(Note: PICS/PIXIT settings are captured at the beginning of the TLI log)

In the log file (in .txt format) the complete test case execution can be seen. All TLI events are presented and message contents are fully decoded and can be verified. Preliminary verdicts and the final test case verdict can be seen in the log file.
References

	[1]
	R5s120892:    Supporting information for agreement of TC 13.4.4.1. This archive comprises:

                        - text format execution log files

